
Příloha: Obecné poznámky pro užití dále uvedených sad příkazů 

 

1. Dále uvedené sady příkazů (str. 2 MAP a str. 3–4 paralelní analýza) se v prostředí SPSS 
nakopírují do okna, které má název syntax. Jeho otevření provedeme volbou File – Open – 
Syntax. 

2. Komentáře (jsou vždy uvedeny hvězdičkou) ukazují uživateli, jaká nastavení je třeba 
provést pro správný výpočet odpovídající vašim datům. 

3. Poté, co provedete úpravy příkazů, vyberete vše (myší, klávesnicí či zkratkou Ctrl+A) a 
sadu příkazů spustíte (ikona zelené šipky na liště nebo v nabídce Run – All, případně 
zkratka Ctrl+R). 

4. Ve výstupním okně se objeví výsledky výpočtů a dle nich rozhodnete o počtu faktorů. 

Doporučení: Vždy používejte jen jednu sadu příkazů, tj. buď MAP, nebo paralelní analýzu. 

Zdroj: O’Connor, B. P. 2000. „SPSS and SAS Programs for Determining the Number of 
Components Using Parallel Analysis and Velicer’s MAP Test.“ Behavior Research Methods, 
Instrumentation, and Computers 32: 396–402. 

Upravil a počeštil: Petr Soukup (soukup@fsv.cuni.cz) 

*SPSS skript od O’Connora pro MAP test – počeštěno a doplněny komentáře k práci. 

*Stačí nakopírovat vše od dalšího řádku a dle komentářů upravit, musíte mít otevřený 
datový soubor s proměnnými, pro který chcete MAP test počítat. 

*Nejdříve je nutné určit, pro jaké proměnné se má MAP test (faktorová analýza) počítat, je 
nutno nahradit seznam „var1 to var 25“ seznamem vašich proměnných a dále je třeba určit 
disk a adresář, kam se bude zapisovat korelační matice, tj. nahradit všude dále frázi 
„C:\data.cor“. Dejte pozor na to, že některé disky či adresáře mohou být na vašem počítači 
chráněny proti zápisu. 

correlation var1 to var25 / matrix out ('C:\data.cor') / missing = listwise.  
factor var= var1 to var25 / matrix out (cor = 'C:\data.cor'). 
matrix. 
mget /type= corr /file='C:\data.cor' .  
call eigen (cr,eigvect,eigval). 
compute loadings = eigvect * sqrt(mdiag(eigval)).  
compute fm = make(nrow(cr),2,-9999). 
compute fm(1,2) = (mssq(cr) - ncol(cr)) / (ncol(cr)*(ncol(cr)-1)).  
loop #m = 1 to ncol(cr) - 1. 
compute a = loadings(:,1:#m).  
compute partcov = cr - (a * t(a)). 
compute d = mdiag( 1 / (sqrt(diag(partcov))) ).  
compute pr = d * partcov * d. 
compute fm(#m+1,2) = (mssq(pr) - ncol(cr)) / (ncol(cr)*(ncol(cr)-1)).  
end loop. 

* Tato část skriptu slouží k určení nejmenších hodnot a díky tomu určení počtu faktorů. 
compute minfm = fm(1,2).  
compute nfactors = 0.  
loop #s = 1 to nrow(fm).  
compute fm(#s,1) = #s -1.  
do if ( fm(#s,2) < minfm ). 
compute minfm = fm(#s,2).  
compute nfactors = #s - 1. 
end if.  
end loop. 
print eigval /title="Eigenvalues". 
print fm /title="Velicer’s Average Squared Correlations". 
print minfm /title="The smallest average squared correlation is".  
print nfactors /title="The number of components is". 
end matrix. 

*SPSS skript od O’Connora pro paralelní analýzu – počeštěno a doplněny komentáře k 
práci. 

*Stačí nakopírovat vše od dalšího řádku a dle komentářů upravit, není třeba mít 
otevřený žádný datový soubor. 

set mxloops=9000 length=none printback=none width=80 seed = 1953125. matrix. 

* Na dalších řádcích musíte zadat: 
Ncases – vepište počet případů, kolik máte v datové matici (defaultní nastavení je 500) 
Nvars- vepište počet proměnných, se kterými chcete vstoupit do faktorové analýzy 
(defaultní nastavení je 50) 
Ndatasets – netřeba měnit nastavení (1000), jde o počet generovaných výběrů z nezávislých 
dat, která se užijí pro výpočet paralelní analýzy 
percent – také netřeba měnit nastavení (95), nejčastěji se pro vyhodnocení paralelní analýzy 
používá 95% percentil. 

compute Ncases = 500.  
compute Nvars = 50.  
compute Ndatsets = 1000.  
compute percent = 95. 

* Následující sekce dle nastavení z předchozí sekce počítá korelační matice a jejich vlastní 
(jednotková) čísla. 
compute evals = make(nvars,ndatsets,-9999).  
compute nm1 = 1 / (ncases-1). 
loop #nds = 1 to ndatsets. 
compute x = sqrt(2 * (ln(uniform(ncases,nvars)) * -1) ) &*  
cos(6.283185 * uniform(ncases,nvars) ). 
compute vcv = nm1 * (sscp(x) - ((t(csum(x))*csum(x))/ncases)).  
compute d = inv(mdiag(sqrt(diag(vcv)))). 
compute evals(:,#nds) = eval(d * vcv * d).  
end loop. 

* Následující sekce slouží k výpočtu percentilů z hodnot vlastních čísel, která se užijí k 
vyhodnocení paralelní analýzy. 
compute num = rnd((percent*ndatsets)/100). 
compute results = { t(1:nvars), t(1:nvars), t(1:nvars) }.  
loop #root = 1 to nvars. 
compute ranks = rnkorder(evals(#root,:)).  
loop #col = 1 to ndatsets. 
do if (ranks(1,#col) = num). 
compute results(#root,3) = evals(#root,#col).  
break. 
end if.  
end loop.  
end loop. 

compute results(:,2) = rsum(evals) / ndatsets.  
compute specifs = {ncases; nvars; ndatsets; percent}.  
print specifs /title="Specifications for this Run:" 
/rlabels="Ncases" "Nvars" "Ndatsets" "Percent".  
print results /title="Random Data Eigenvalues" 
/clabels="Poradi" "Prumer" "95% Perc".  
end matrix. 

* Práce s výsledky je následující: Hodnoty ze sloupce 95% Perc porovnáváte s hodnotami 
vlastních čísel (eignevalues), která vám poskytne výpočet faktorové analýzy v SPSS. 
* Vybíráte tolik faktorů, pro kolik jich platí, že vlastní čísla z výpočtu FA z vašich dat jsou 
větší než 95% percentily nagenerované paralelní analýzou. 
* Příklad užití najdete v češtině v učebnici Rabušic, Soukup, Mareš (2019). 

